# $Id$
# Maintainer: Jan de Groot <jgc@archlinux.org>

pkgname=xorg-xwd
pkgver=1.0.7
pkgrel=1
pkgdesc="X Window System image dumping utility"
arch=('x86_64')
url="https://xorg.freedesktop.org/"
license=('custom')
depends=('libxkbfile')
makedepends=('xorg-util-macros' 'libxt')
groups=('xorg-apps' 'xorg')
source=(https://xorg.freedesktop.org/archive/individual/app/xwd-${pkgver}.tar.bz2{,.sig})
sha512sums=('6fdb0a115ae505169bbc7fcc45580daf26bced2e26e0bf9a877636016038931a8386e8edcbe015f4024be508d0461906adef4834cf22a445e3c282f0b8fad17e'
            'SKIP')
validpgpkeys=('4A193C06D35E7C670FA4EF0BA2FB9E081F2D130E') # "Alan Coopersmith <alan.coopersmith@oracle.com>"

build() {
  cd xwd-${pkgver}
  ./configure --prefix=/usr
  make
}

package() {
  cd xwd-${pkgver}
  make DESTDIR="${pkgdir}" install
  install -Dm644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/COPYING"
}
